Numerical Models

Numerical calculations are essential for the solution of the problems of everyday life. These problems often lead to such mathematical models the solution of which cannot be determined by closed mathematical formulas in most cases. However, we have to keep trying. The usage of approximation methods provides an alternative solution in these cases.

We can do the total examination of the Lorenz system of differential equation without knowing its analytic solution. The Statistics package belongs to the latest version of Maple. We are going to show its usage with the help of a simple independence examination exercise. We are going to discuss the numerical solution of nonlinear system of equations. At the end of this chapter, we are going to show the results of the Maple-NAG (Numeric Algorithm Group) integration that can be achieved at present.

8.1 The Chaotic Behaviour of the Lorenz Equations

Determine the solutions that satisfy the initial condition of the Lorenz system of differential equation.

$$\frac{d}{dt} x(t) = 10 \cdot (y(t) - x(t))$$

$$\frac{d}{dt} y(t) = 28 \cdot x(t) - x(t) \cdot z(t) - y(t)$$

$$\frac{d}{dt} z(t) = x(t) \cdot y(t) - \frac{8 \cdot z(t)}{3}$$

$$x(0) = 0, y(0) = 1, z(0) = 0$$

Prove that the local maximums of the z(t) coordinate of the solution behave chaotically.

It was Edward Lorenz who published this system of differential equation in 1963 when he published his researches concerning the modelling of heat flow of liquids. Lorenz modelled the flow of a liquid layer heated from beneath and cooled from above. You can see this kind of liquid movement when you boil water. The liquid does a very complex vertical, up and down flowing, whirling movement. In the differential equation the x denotes the speed of the movement and the y and z are the temperature of the liquid. The system of differential equation seemed so simple that nobody believed that it would be able to describe this complex phenomenon.

The constants in the system of differential equation are given by the physical properties of the liquid and the thickness of the layer. By having chosen these constants, Lorenz stated that the solutions behave in a chaotic way. We really hope that by reaching the end of this worksheet we will be able to raise your interest so that you can examine the chaotic behaviour more carefully.

Let’s create the Lorenz differential equations in a three-element list. Put the three initial conditions in a set data structure.

> $restart$

> $Lorenz := \left[(\mathrm{D}(x))(t) = 10\,y(t) - 10\,x(t),\ (\mathrm{D}(y))(t) = 28\,x(t) - x(t)\,z(t) - y(t),\ (\mathrm{D}(z))(t) = x(t)\,y(t) - \frac{8\,z(t)}{3} \right]$

$$Lorenz := \left[(\mathrm{D}(x))(t) = 10\,y(t) - 10\,x(t),\ (\mathrm{D}(y))(t) = 28\,x(t) - x(t)\,z(t) - y(t),\ (\mathrm{D}(z))(t) = x(t)\,y(t) - \frac{8}{3}\,z(t) \right] \tag{1}$$

> $kezdeti := \{x(0) = 0, y(0) = 1, z(0) = 0\}$

$$kezdeti := \{x(0) = 0, y(0) = 1, z(0) = 0\} \tag{2}$$

We will try to determine the solutions of the Lorenz equations with the help of the dsolve procedure and built-in functions for the initial conditions.

> $dsolve(convert(Lorenz, set) \cup kezdeti,\ \{y(t), x(t), z(t)\})$

We have not received any responds, that is, the reply of the system is NULL, an empty series. This does not mean that there is no solution for it. It only means that there is no closed formula for it or if there is then Maple is unable to create it. However, to draw the solution we do not need to create the solution itself. The DEplot3d of the DETools package plots the numerical solution without determining the solution with a closed formula.

So that the graph to be created should not be fragmented we can regulate the distance between the points of the net the points of which is used by the system to generate the graph by the stepsize option of the procedure. It is not easy to set the value of the stepsize option well. After several tries we have decided to set the value of stepsize to 0.005. The t interval of the representation is given as [0,50]. The colour of the curve can be changed in the linecolor option according to the value of the sine function.

> $with(DEtools):$

> $DEplot3d(Lorenz,\ [x(t), y(t), z(t)],\ t = 0..50,\ [[op(kezdeti)]],\ stepsize = 0.005,\ scene = [x(t), y(t), z(t)],\ linecolor = \sin(t),\ orientation = [-50, 100])$



The 3-D curve, or trajectory, created plots the so-called strange attractor. The solution seems to move regularly around two ellipse-shaped figures in the rectangular shaped domain.

$$T = [\ -20 < x < 15,\ -20 < y < 20\ ,\ 0 < z < 50\ ]$$

The ellipses do not let the curve go far away which plots two irregular surfaces that connect to each other in a V shape. The solution moves around one of the ellipses then around the other one. We can be sure of this by rotating the 3-D graph.

After having finished this visual illustration let's get down to the numerical solution of the task which we can do with the numeric option of the dsolve procedure. The output=listprocedure option used in the instruction guarantees that the x(t), y(t) and z(t) coordinates of the solution can be received in a procedure. The procedures provide the further calculation of the values of the solutions.

```
> mo := dsolve(convert(Lorenz, set) ∪ kezdeti, {y(t), x(t), z(t)}, numeric, output = listprocedure)
mo := [t = proc(t) ... end proc, x(t) = proc(t) ... end proc, y(t) = proc(t) ... end proc, z(t) =     (3)
    proc(t)
    ...
 end proc]
```

Introduce the X, Y and Z for the x(t), y(t) and z(t) procedures.

> $X := subs(mo, x(t))$

$$X := \textbf{proc}(t)\ ...\ \textbf{end proc} \qquad (4)$$

> $Y := subs(mo, y(t))$

$$Y := \textbf{proc}(t)\ ...\ \textbf{end proc} \qquad (5)$$

> $Z := subs(mo, z(t))$

$$Z := \textbf{proc}(t)\ ...\ \textbf{end proc} \qquad (6)$$

From now on, if we want to calculate the x,y,z coordinates of the solution, e.g. at the place of t=10 we only have to enter the [képlet] instruction.

> $[X(10), Y(10), Z(10)]$

$$[-5.91637069204104104, -5.52312124760978840, 24.5722252164589962] \qquad (7)$$

We can see that the procedures created as the solutions of dsolve generate the substitution values of the curve to 18 significant digits although we did not ask for it. Check the value of the Digits environment variable.

Before determining the maximums of the Z function, it is useful to plot the graph of the z(t) function in the [10,50] interval.

> $plot(Z(t), t = 10..50)$



The graph of the Z(t) function in the [10,50] time interval flows with changing intensity. It looks like an intensity diagram of live speech. Although the graph is not suitable for determining the exact approximation of the maximum locations we can still see that the function has 54 maximum locations (we have counted it) in the interval examined. Its lower limit is z=7 and its upper limit is z=44.

The examination of the local maximums of the Z(t) function was first mentioned by E. Lorenz. He concentrated on the assumption that the solutions behaved in a chaotic way which he wanted to prove. For the time being, the chaotic behaviour means that the solution meanders helter-skelter in the rectangular and visits even the smallest parts infinite times. Thus it cannot stay at any of the small cubes. Why?

It meanders not like the sine function which gets every value between [-1,1] periodically and infinite times. The meanders of the sine are regular while the chaotic meander is unpredictable. Lorenz's idea is creative because he suggested that the helter-skelter meander should point to the maximum locations of the function instead of the whole of it. If the maximums behave in a chaotic way, that is, they are unpredictable then the whole curve behaves chaotically.

This argument was acceptable for everybody. However, we cannot examine infinite numbers but finite numbers of maximum locations. Thus we have to draw general deductions with the help of the finite amount of maximum locations concerning the other maximum locations.

Let's look for the maximums. For this, it is useful to know the first and second derivatives of the z(t). We need the zero locations of the first derivative and the signs of the second derivative at the critical points. We already have the first derivative of the z(t) because this is exactly the third equation of the Lorenz system of differential equation.

$$\frac{d}{dt} z(t) = x(t)\, y(t) - \frac{8\, z(t)}{3}.$$

In this formula we can calculate the values of the x(t), y(t) and z(t) functions because they are given by the X(t), Y(t) and Z(t) procedures at an arbitrary t point. Thus we can consider as if we had the $\frac{d}{dt} z(t)$ derivative at the arbitrary t point.

We can find the zero locations of the derivative in an interval with the fsolve command. The first parameter of the fsolve command is the $x(t)\, y(t) - \frac{8\, z(t)}{3} = 0$ equation which refers to the zero locations of the derivative. Its second parameter is the t time interval where we are looking for the solutions of the equation. We can see in the graph that the z(t) has a maximum and a minimum location in the [10,11] interval. Let's look for the zero locations of the derivative in the [10,11] interval.

> $T_1 := fsolve\left(X(t)\, Y(t) - \frac{8\, Z(t)}{3} = 0,\, t = 10\,..11\right)$

$$T_1 := 10.41236461 \qquad (8)$$

Unfortunately, the fsolve has returned only one out of the two zero locations. And it has given the lower one. To get the zero location larger than 10.41236461 we have to change the interval of the search. Let's try the [10,5,11] interval.

> $T_2 := fsolve\left(X(t)\, Y(t) - \frac{8\, Z(t)}{3} = 0,\, t = 10.5\,..11\right)$

$$T_2 := 10.76475475 \qquad (9)$$

>

So we have two critical points but it is unknown where the z(t) function has a maximum and a minimum out of the T1 and T2 locations. For this, we should be able to determine the beginning of the second derivative. We do not have the explicit formula of the $\frac{\partial^2}{\partial t^2} z(t)$ second derivative. However, we can calculate it by differentiating the Lorenz equation.

> $d2z := \frac{\partial}{\partial t}\, rhs(Lorenz_3)$

$$d2z := \left(\frac{d}{dt} x(t)\right) y(t) + x(t) \left(\frac{d}{dt} y(t)\right) - \frac{8}{3}\left(\frac{d}{dt} z(t)\right) \qquad (10)$$

Let's substitute the right sides of the Lorenz equations in the d2z expression into the places of the $\frac{d}{dt} x(t)$, $\frac{d}{dt} y(t)$ és $\frac{d}{dt} z(t)$ derivatives.

> $map(convert, Lorenz, diff)$

$$\left[\frac{d}{dt} x(t) = 10\, y(t) - 10\, x(t),\ \frac{d}{dt} y(t) = 28\, x(t) - x(t)\, z(t) - y(t),\ \frac{d}{dt} z(t) = x(t)\, y(t) - \frac{8}{3} z \qquad (11)$$

$(t)]$

> $expand(subs(\%, d2z))$

$$10\, y(t)^2 - \frac{41}{3}\, x(t)\, y(t) + 28\, x(t)^2 - x(t)^2\, z(t) + \frac{64}{9}\, z(t) \qquad \textbf{(12)}$$

With this calculation we have received an expression which shows they way the $\frac{\partial^2}{\partial t^2}\, z(t)$ can be calculated from the x(t), y(t) and z(t) functions. This sounds good although we do not know these functions. However, we have the procedures which give the numerical approximation of the value of each function for every t. Thus it is obvious that if we change the functions in the (12) expression to procedures then we get a new procedure which is able to calculate the substitution values of the second derivative.

> $D2Z := unapply\left(10\, Y(t)^2 - \frac{41\, X(t)\, Y(t)}{3} + 28\, X(t)^2 - X(t)^2\, Z(t) + \frac{64\, Z(t)}{9}, t\right)$

$$D2Z := t \rightarrow 10\, Y(t)^2 - \frac{41}{3}\, X(t)\, Y(t) + 28\, X(t)^2 - X(t)^2\, Z(t) + \frac{64}{9}\, Z(t) \qquad \textbf{(13)}$$

Let's use the D2Z procedure to decide if we are going to find an extremum at the T1 and T2 critical points previously calculated. If yes then what kind of extremum is it?

> $('D2Z')(T_1) = D2Z(T_1), ('D2Z')(T_2) = D2Z(T_2)$

$$D2Z(10.41236461) = -649.1014625, D2Z(10.76475475) = 386.8395318 \qquad \textbf{(14)}$$

In the first case the second derivative is negative which means that the z(t) function has a maximum at the T1 point. The second derivative is positive at the T2 point so we will find a minimum here. A question can arouse at this point: would not it have been easier to solve the maximum-minimum problem by comparing the two function values, like this:

> $'Z'(T_1) = Z(T_1), 'Z'(T2) = Z(T_2);$

$$Z(10.41236461) = 31.6648066678655270, Z(T2) = 21.7559109880570994 \qquad \textbf{(15)}$$

We can see that we have found a bigger function value at the T1 point which coincides with the negative sign of the second derivative. But be careful since a local maximum can be smaller than a local minimum. Let's keep on examining the sign of the second derivative.

So we have a method to calculate the maximum of the z(t) function in an interval which contains a critical point. How can we use this to find all the local maximums of the z(t) function in the [10,50] interval?

Let's start from the t=10 start point and step by [képlet] steps. To choose the right step we should keep in mind that it should not be too big for fear of crossing the maximum place. However, if we choose a too small step value then Maple has to do lots of examinations so our procedure will be slow. The other disadvantage is that there will be empty intervals where the derivative does not have a zero location. We have to be prepared to tackle these issues. Let's see what the fsolve returns for the [10.5, 10.6] interval.

> $fsolve\left(X(t)\, Y(t) - \frac{8\, Z(t)}{3} = 0, t = 10.5\,..10.6\right)$

(16)

$$fsolve\left(X(t)\,Y(t) - \frac{8}{3}\,Z(t) = 0,\, t,\, 10.5\,..10.6\right) \qquad \textbf{(16)}$$

So if there is no zero location in the interval examined then the fsolve repeats the command entered. And in case it finds a root it returns a floating point value.

> $whattype(T_1)$

$$float \qquad \textbf{(17)}$$

According to this, the whattype procedure can help to decide if there is a root in the interval examined. If there is a root then the type of the output of the fsolve is going to be float otherwise not.

After this we are going to examine if the derivative of the Z(t) has a zero location in certain sub intervals with the help of a for cycle containing 801 steps. If the content of the T variable received is float then we have to look at if the second derivative of the Z(t) is negative at the T. If both conditions are true then we concatenate the value of the Z(T) to the P sequence. We are going to calculate from the t=10 start value to the t=10+800.0.05=50 value.

```
> P := NULL: t0:=10: dt:=0.5e-1: N:= 800:
```

```
> for k from 0 to N do
   T:=fsolve(X(t)*Y(t)-8/3*Z(t),t=t0+k*dt..t0+(k+1)*dt);
   if whattype(T)=float and D2Z(T)<0 then
    P := P, Z(T)
   fi
  od:
```

> $P := [P];$

> $Maximum\ helyek\ szama = nops(P)$

$P :=$ [31.6648066678655270, 32.0287240281795889, 32.4392437247626546,
32.9084271994090970, 33.4539624892811034, 34.1033502053308268,
34.9029996852020972, 35.9413941827897930, 37.4261532543832019,
40.1439576863332022, 39.1685109170936343, 41.5346165741923202,
36.8791794033177496, 39.0067853056162974, 42.1154136499328118,
36.0649750726815199, 37.6165350302293718, 40.6081695240345796,
38.3283390958434325, 43.2333139170710084, 34.6187287620917914,
35.5623781528252040, 36.8610179644865070, 38.9736699077084694,
42.2497815733780158, 35.8838432364849070, 37.3369593277591676,
39.9413904822739240, 39.5733449524780242, 40.3988526330474400,
38.6944947383301782, 43.8351296583691124, 33.8846292433924674,
34.6278829630941374, 35.5761322539704352, 36.8809937702023092,
39.0112428075655799, 42.0978027964608614, 36.0889220084196794,
37.6541424160721334, 40.7059903991933966, 38.1632841384804280,
42.3818022506712211, 35.7080816445651266, 37.0732223172757856,
39.3850089761126228, 40.8861022923321614, 37.8681414382105786,
41.3128018097937329, 37.2062010801167276, 39.6573703185453540,

40.1974885490067777, 39.0661170200630963, 41.8898262201479882]

Maximum helyek szama = 54 **(18)**

It is obvious that the P list contains only maximum locations. But does it contain all of them? Since we can see in the graph of the z(t); that the function has 54 maximum locations and the P contains 54 elements as well we can state that the P contains all the local maximums of the z(t) function in [10,50] interval.

To examine the chaotic behaviour of the maximums in the P we have to generate the [képlet] pairs in the case of k=1,2,...53 then we have to plot the points got in the plane. We are looking forward to the result of this interesting construction. We can create the pairs from the elements of the P list with the seq instruction. This list consisting of two-element lists is a suitable input for the plotting of the point sequence.

> $parositas := [seq([P_k, P_{k+1}], k = 1..nops(P) - 1)]:$
$pontok := plot(parositas, style = point, symbol = cross, title = A\ Lorenz\text{-}egyenlet\ z\text{-}koordinatainak\ egymast\ koveto\ maximumai, scaling = constrained):$
$pontok;$



>

This shape is similar to a yurt. In the middle, there is a pole supporting the cover which is stretched at the edges. But due to the weight of the cover it sags a little. What kind of function describes this curve?

We can consider the curve fitting methods, e.g. interpolation, spline, the minimax and the method of the least square. We have to choose the appropriate fitting method. Obviously we should choose the method of the least square.

Choose the method of the least square and fit a parabola to the data sequence on the right and left side. But before this we have to sort the point pairs. By clicking on the graph above we can check that the imaginary limit is around the 38.5 value of the horizontal axis. After several checks we have chosen the mean as 38.63667. Let's sort the points in one cycle in a way that if the first coordinate is smaller than 38:63667 then we put it on the left side otherwise on the right side.

```
> fele := 38.63667;bal := NULL:jobb := NULL:
  for k to nops(parositas) do
   if parositas[k][1] < fele then
     bal := bal, parositas[k]
   else jobb := jobb, parositas[k]
   end if end do;
  bal := [bal];`Bal oldali pontok száma` = nops(bal);
  jobb := [jobb]; `Jobb oldali pontok száma` = nops(jobb)
```

fele := 38.63667

bal := [[31.6648066678655270, 32.0287240281795889], [32.0287240281795889, 32.4392437247626546], [32.4392437247626546, 32.9084271994090970], [32.9084271994090970, 33.4539624892811034], [33.4539624892811034, 34.1033502053308268], [34.1033502053308268, 34.9029996852020972], [34.9029996852020972, 35.9413941827897930], [35.9413941827897930, 37.4261532543832019], [37.4261532543832019, 40.1439576863332022], [36.8791794033177496, 39.0067853056162974], [36.0649750726815199, 37.6165350302293718], [37.6165350302293718, 40.6081695240345796], [38.3283390958434325, 43.2333139170710084], [34.6187287620917914, 35.5623781528252040], [35.5623781528252040, 36.8610179644865070], [36.8610179644865070, 38.9736699077084694], [35.8838432364849070, 37.3369593277591676], [37.3369593277591676, 39.9413904822739240], [33.8846292433924674, 34.6278829630941374], [34.6278829630941374, 35.5761322539704352], [35.5761322539704352, 36.8809937702023092], [36.8809937702023092, 39.0112428075655799], [36.0889220084196794, 37.6541424160721334], [37.6541424160721334, 40.7059903991933966], [38.1632841384804280, 42.3818022506712211], [35.7080816445651266, 37.0732223172757856], [37.0732223172757856, 39.3850089761126228], [37.8681414382105786, 41.3128018097937329], [37.2062010801167276, 39.6573703185453540]]

Bal oldali pontok száma = 29

jobb := [[40.1439576863332022, 39.1685109170936343], [39.1685109170936343, 41.5346165741923202], [41.5346165741923202, 36.8791794033177496], [39.0067853056162974, 42.1154136499328118], [42.1154136499328118, 36.0649750726815199], [40.6081695240345796, 38.3283390958434325], [43.2333139170710084, 34.6187287620917914], [38.9736699077084694, 42.2497815733780158], [42.2497815733780158, 35.8838432364849070], [39.9413904822739240, 39.5733449524780242], [39.5733449524780242, 40.3988526330474400], [40.3988526330474400, 38.6944947383301782], [38.6944947383301782, 43.8351296583691124], [43.8351296583691124, 33.8846292433924674], [39.0112428075655799, 42.0978027964608614], [42.0978027964608614, 36.0889220084196794], [40.7059903991933966, 38.1632841384804280], [42.3818022506712211, 35.7080816445651266], [39.3850089761126228, 40.8861022923321614], [40.8861022923321614, 37.8681414382105786], [41.3128018097937329, 37.2062010801167276], [39.6573703185453540, 40.1974885490067777], [40.1974885490067777, 39.0661170200630963], [39.0661170200630963, 41.8898262201479882]]

Jobb oldali pontok száma = 24 **(19)**

There are 29 results on the left and 24 on the right side. We can get the parabola that can be fitted by the method of the least square with the PolynomialFit procedure. This procedure is in the Statistics package. It requires the x and y coordinates of the points as parameters in a separate list. So we have to sort the x and y coordinates of the left point sequence.

```
> with(Statistics):
  Xb :=[seq(bal[k][1], k = 1 ..nops(bal))]:
  Yb :=[seq(bal[k][2], k = 1 ..nops(bal))]:
  fb := PolynomialFit(2, Xb, Yb, x);
```

$$fb := 142.879722024536932 - 7.62141514311384949\, x + 0.130447658424413498\, x^2 \quad \textbf{(20)}$$

The first parameter of the PolynomialFit procedure is the degree of the approximation polynomial, that is, 2. The second and the third parameters are the lists of the x and y coordinates of the points. The fourth parameter is the name of the variable of the polynomial to be created. So we have received that the value of the fb variable is the parabola that matches the left side point sequence.

We match the parabola onto the right side points the same way.

```
> Xj :=[seq(jobb[k, 1], k = 1 ..nops(jobb))]:
  Yj :=[seq(jobb[k, 2], k = 1 ..nops(jobb))]:
  fj := PolynomialFit(2, Xj, Yj, x);
```

$$fj := 483.585322541109917 - 19.9371169216980632\, x + 0.220970722553671589\, x^2 \quad \textbf{(21)}$$

With the help of the fb and fj parabolas we can create the function defined by T sections. We can do this with the piecewise procedure.

```
> T := unapply(piecewise(x < fele, fb, fj), x):'T'(x)= T(x);
```

$$T(x) = \begin{cases} 142.879722024536932 - 7.62141514311384949\, x + 0.130447658424413498\, x^2 & x < 38.63667 \\ 483.585322541109917 - 19.9371169216980632\, x + 0.220970722553671589\, x^2 & \text{otherwise} \end{cases}$$

(

Finally, we can display the T(x) curve and points in the same coordinate system.

```
> grafT := plot(T(x), x = 31 ..45):
  unimodal := plots[display]([grafT, pontok], scaling = constrained): unimodal;
```

A Lorenz-egyenlet z-koordinatainak egymast koveto maximumai



Notice that the fitted curves follow the curve of the points very well. We want to highlight that the two parabolas continuously match at the connection point. Let's see.

```
> csucs := fsolve(fb = fj, x = 32 ..44);
  subs(x = fele, fb) = subs(x = fele, fj); eltérés = rhs(%)-lhs(%);
```

$$csucs := 38.63667477$$

$$43.1448762 = 43.1449017$$

$$eltérés = 0.0000255 \quad (23)$$

The difference is within the 10(-4) tolerance.

The representation of the chaotic behaviour of the maximums is still ahead. To prove it, the T(x) continuous mapping is highly suitable. If we begin with the z1 start value and apply the T mapping on it more times then we can get a recursive sequence

$$z_2 = T(z_1),\ z_3 = T(z_2), z_4 = T(z_3) ,...,\ z_{n+1} = T(z_n) , ...$$

for which the $[z_n, z_{n+1}]$ points match the curve.

In theory, the first 54 elements of this zn series should be the same as the maximum value above for the zn series. However, it is not perfect. We can assume that this will be true for all the other subsequent maximums.

Let’s check the first 54 elements. Apply the T mapping for the z1 starting point k times. Thus we get the kth iteration of the z point by the T mapping. We can give the k times composition of the T mapping with the T@@k formula.

```
> kezdo := P[1]; palya :=[kezdo, seq( (T @@ k)(kezdo), k = 1 ..53 ) ];
```

kezdo := 31.6648066678655270

palya := [31.6648066678655270, 32.3437316, 32.8382259, 33.2740823, 33.7111507, 34.1992039, 34.8030874, 35.6363092, 36.9421669, 39.3531422, 41.2078103, 37.2471573, 39.9803270, 39.6983377, 40.3555038, 38.8784825, 42.4657622, 35.4260112, 36.5954912, 38.6697056, 43.0505958, 34.8174888, 35.6573412, 36.9774728, 39.4245036, 41.0272971, 37.5658651, 40.6616506, 38.2555664, 42.2267987, 35.7181538, 37.0802067, 39.6340022, 40.5103588, 38.5582223, 42.9527987, 34.9087221, 35.7918380, 37.2059779, 39.8942283, 39.8952609, 39.8928796, 39.8983716, 39.8857093, 39.9149233, 39.8476286, 40.0032091, 39.6465532, 40.4800052, 38.6201625, 43.1043246, 34.7691649, 35.5869817, 36.8598150] **(24)**

Compare the trajectory of the z1 point by the T mapping with the maximum values received earlier. Represent the difference between the two lists.

```
> plot( [seq( [k, palya[k]-P[k]], k = 1 ..54 ) ], title = `Elteresek`);
```



This is the first time we have met the chaos. Naturally we have thought that we would not get back the maximums exactly but we did not except such big differences. This leads us to a more precise definition of the chaos. Namely if we start the iteration from two points which are at an arbitrary distance from each other, then some iterations will differ from each other by larger values than a positive [képlet] previously given. This is called the sensitive dependence. Let's see the arbitrary meander of the iterations.

```
> identitas := plot(x, x = 31 ..45, color = green);
  orbit := seq(plot([[palya[k], palya[k]], [palya[k], palya[k+1]], [palya[k+1], palya[k+1]]], linestyle = 3), k = 1 ..53);
  dinamika := plots[display]([unimodal, orbit, identitas], title = Egy pont palyaja); dinamika;
```



The trajectory has points which are very close to each other and they go near each other then later they start to leave each other. This graph raised the problem of finding enough conditions for the chaotic behaviour. The article titled “Period Three Implies Chaos” by T.Y. Li and J.A. Yorke published in 1975 refers to the simple conditions that guarantee the chaos. More precisely, if the [képlet] function has a p point with three periods, that is,

$f(p) = q, f(q) = r, f(r) = p;$ másképpen $f(f(f(p))) = p$;**then** *the f mapping is chaotic.*
To prove that our T mapping is chaotic we only have **to** *show that the T has a cycle with three periods. As a first step let's look* **for** *such an* $[a, b]$ *interval which the T mapping transfers into itself. The b point is the value of the maximum of the T function* **and** $a = T(b)$.

```
> b := subs(x = csucs, fb); a := T(b);
  negyzet := plot([[a, a], [b, a], [b, b], [a, b], [a, a]], color = blue, thickness = 2) : plots_display(
      [negyzet, dinamika], title = Diszkret dinamikus rendszer);
```

$b := 43.1448878$

$a := 34.7335274$



The image of the [a,b] interval is obviously a part of the [a,b] interval by the T mapping. This means that the trajectory of the z1 point originating from the [a,b] interval stays at the [a,b] interval. We have created the frames of the discrete dynamic system. Let's check if we can find a p point in the [a,b] interval which is the point of the T mapping with three periods. Let's create the threefold composition of the T piecewise function with itself.

```
> Digits := 18 : iteralt3 := simplify(T(T(T(x))));
```

$$iteralt3 := \begin{cases} 9.75345795718418092\ 10^{5} - 3.05747150240608740\ 10^{5}\ x + 41112.2762373092240\ x^{2} - 3096.15\ldots \\ 5.84544329152944893\ 10^{5} - 1.81860520657280043\ 10^{5}\ x + 24346.8090298833455\ x^{2} - 1829.60\ldots \\ 9.75345795718418092\ 10^{5} - 3.05747150240608740\ 10^{5}\ x + 41112.2762373092240\ x^{2} - 3096.15\ldots \\ 6.17672788820095171\ 10^{5} - 1.66994312147417767\ 10^{5}\ x + 19744.5556335592028\ x^{2} - 1334.89\ldots \\ 3.71616224290259755\ 10^{5} - 99520.8375704729817\ x + 11703.4625791019816\ x^{2} - 789.1187\ldots \\ 6.17672788820095171\ 10^{5} - 1.66994312147417767\ 10^{5}\ x + 19744.5556335592028\ x^{2} - 1334.89\ldots \\ 9.75345795718418092\ 10^{5} - 3.05747150240608740\ 10^{5}\ x + 41112.2762373092240\ x^{2} - 3096.15\ldots \\ 5.84544329152944893\ 10^{5} - 1.81860520657280043\ 10^{5}\ x + 24346.8090298833455\ x^{2} - 1829.60\ldots \\ 2.35489538612441180\ 10^{8} - 4.28256153686833477\ 10^{7}\ x + 3.39532502008972780\ 10^{6}\ x^{2} - 1.5327\ldots \\ 3.98607016139644280\ 10^{8} - 7.25169461577290267\ 10^{7}\ x + 5.75056440493526700\ 10^{6}\ x^{2} - 2.596\ldots \\ 1.60110349813799670\ 10^{8} - 2.81156396094851908\ 10^{7}\ x + 2.16273837670606689\ 10^{6}\ x^{2} - 9519\ldots \\ 9.46311073532072666\ 10^{7} - 1.66075655484960370\ 10^{7}\ x + 1.27707279063774698\ 10^{6}\ x^{2} - 5620\ldots \\ 1.60110349813799670\ 10^{8} - 2.81156396094851908\ 10^{7}\ x + 2.16273837670606689\ 10^{6}\ x^{2} - 9519\ldots \\ 3.98607016139644280\ 10^{8} - 7.25169461577290267\ 10^{7}\ x + 5.75056440493526700\ 10^{6}\ x^{2} - 2.596\ldots \\ 2.35489538612441180\ 10^{8} - 4.28256153686833477\ 10^{7}\ x + 3.39532502008972780\ 10^{6}\ x^{2} - 1.5327\ldots \\ 3.98607016139644280\ 10^{8} - 7.25169461577290267\ 10^{7}\ x + 5.75056440493526700\ 10^{6}\ x^{2} - 2.596\ldots \end{cases}$$

While we could create the formula of the T from two parts, the formula of the [képlet] consists of (22)2 =16 intervals and the formulas considered at certain sub intervals are maximum 8th degree polynomials. The x=p solution of the [képlet] equation is going to be a point with three periods of the T. Let’s denote with q the value of the T(p), with r the T(q). Finally, the T(r) has to return the p initial value.

```
> p := fsolve(iteralt3 = x, x = 32 ..44); q := T(p); r := T(q); T(r)-p;
```

$$p := 38.0789150426094938$$

$$q := 41.814098895782417$$

$$r := 36.282124129973632$$

$$-3.155398\ 10^{-10} \quad \textbf{(26)}$$

The $T(T(T(p))) - p$ difference stays below the 10(-9) so p is really a point with three periods of the T. Plot the three-period trajectory of the p point into the graph of the T function.

```
> identitas := plot(x, x = 31 ..45, color = green) :
  ciklus3 := plot([[p, 0], [p, q], [q, q], [q, r], [r, r], [r, p], [p, p]], linestyle = 3) :
  plots[display]([unimodal, ciklus3, identitas, negyzet], view = [31 ..45, 31 ..45], scaling
      = constrained);
```



The graph beautifully illustrates that the trajectory with three periods exists. According to the Li-Yorke theorem we can come to the conclusion that the T continuous mapping is chaotic indeed.

We are finishing this worksheet with the graph of the

$T(T(T(x)))$ third iterated function into which we plotted the y=x line. We can find the three points of the trajectory with three points between the intersection points of the bisector line and the $T(T(T(x)))$ function. The fix point of the T(x) and another trajectory with three points also create an intersection point though not plotted in the graph. It will be your task to find them.

```
> vonal:= plot([[[p,a],[p,p]],[[q,a],[q,q]],[[r,a],[r r]]],
  linestyle=3,color=black);
  period3:= plot([iteralt3,x],x = a..b,title =`A T(x) harmadik
  iteraltja`);
  plots[display]([vonal, period3]);
```



What Have You Learnt About Maple?

- We can see the 3-D curves of the numerical solutions of the system of differential equations if use the DEplot3d procedure of the DETools package. In this chapter you have become familiar with a simple version of the call of the procedure.

DEplot3d([diffegyenletek] ,[x(t), y(t), z(t)],t = t0..t1,[kezdeti feltételek]], stepsize=Δt, scene=[x(t),y(t),z(t)])

The default numerical method of the calculation to create the solutions is the Runge-Kutta method, also known as rk4.

Besides the system of differential equation and the lists of the initial conditions we also give the
Δt] step value of the t parameter of the curve as a parameter with the stepsize option of the procedure. The step value and the interval of the representation together determine the degree of the difference between the calculated and the “real” curve and it also determines the length of the calculation. We suggest that you should try to set them appropriately according to the given system of differential equation. In the case of a chaotic system the numerical solution can highly differ from the “real” solution in the “capital” t interval if the Δt step value is big.

- • The dsolve procedure is not only used to create the symbolic solutions of the system of differential equation but to give the numerical solutions as well. In the case of this call of the procedure *mo:=dsolve({ diffegyenletek, kezdeti_feltételek }, {változók}, numeric, output = listprocedure)*

we get the numerical solution in the form of procedures. The values of the procedures and the solutions can be created by the mo(t) call at the t place. We want to highlight that the exactness of the calculation is 18 significant digits irrespectively of the settings of the Digits.

- With the help of the whattype procedure we can examine the built-in data type of a given expression. So far you have become familiar with the following data types:

Maple típusnév	***Az adattípus neve***
integer	*egész szám*
float	*lebegőpontos, tizedes tört*
complex	*complex szám*
set	*halmaz*
series	*sorozat*
list	*lista*
+	*összeg*
*	*szorzat*
=	*egyenlet*
Array	*tömb*

- With the help of the PolynomialFit procedure of the Statistics package we can find a polynomial that proceeds the closest to a certain 3-D point system by the method of the least square. Its call is

PolynomialFit(fokszám, X, Y, változó_név);

in which case the degree is the degree of the approximation polynomial, the X and Y are the names of Vector or list type of variable the content of which is the coordinates of the 2-D points. The name of the variable is used by the procedure to create the polynomial. In other words, this name will be the independent variable of the polynomial.

- The k times composition of the y=f(x) function can be given by the (f@@k) formula. Thus in the case of k=1 we get the f function itself and the twice and three times compositions.

- At each section we can create functions interpreted by different formulas if we use the piecewise instruction. Its call is

piecewise(feltétel1, képlet1, feltétel2, képlet2, ..., feltételn, képletn , képletn+1),

If the condition_i is satisfied then the function is determined by the formula_i formula in the case of the i=1,2...n. The last formula is used by Maple to interpret the function in the interval not used so far. If we do not give it then the system uses the default 0. The value returned will be a PIECEWISE Maple object which we consider as an expression which contains an unknown x. Use the unknown x to give the conditions and the formulas. If we want to create a function object from this piecewise object then we can use the unapply instruction of Maple.

Exercises

1. The Rössler attractor. Plot the

$$\frac{d}{dt}x(t) = -y(t) - z(t),\quad \frac{d}{dt}y(t) = x(t) - a\,y(t),\quad \frac{d}{dt}z(t) = b - z(t)\,(x(t) - c)$$

• x(t), y(t), z(t)

• numerical solutions and the,

• a numerikus megoldások $(x(t), y(t), z(t))$ curves of the numerical solutions in 3-D

of the $a = 0.2$, $b = 0.2$ és $c = 5$ Rössler system of differential equation by the $x(0) = 4, y(0) = 0, z(0) = 0$ parameters and the [képlet] initial conditions to the t=200 date. Examine the chaotic behaviour of the solution of the system of differential equation above.

2. Chua-áramkör. Rajzoljuk fel a

$$\frac{d}{dt}x(t) = c_1\,(y(t) - x(t) - p(x(t)))$$

$$\frac{d}{dt}y(t) = c_2\,(x(t) - y(t) + z(t))$$

$$\frac{d}{dt}z(t) = -c_3\,y(t)$$

Chua circuit. Plot

• x(t), y(t), z(t) numerical solutions,

• the $(x(t), y(t), z(t))$ curves of the numerical solutions in 3-D in which case it is

$$p(x) = m_1\,x + \frac{(m_0 - m_1)\,(|x+1| - |x-1|)}{2}$$

and the values of the parameters

$$c_1 = 15.6\,,\ c_2 = 1,\ c_3 = 25.58\,,\ m_0 = -\frac{8}{7},\ m_1 = -\frac{5}{7}\ .$$

Examine the chaotic behaviour of the solution that satisfies the [képlet] initial condition of the system of differential equation. Look for a c3 value for which the chaotic behaviour does not become true. The construction of the Chua circuit is shown in the following figure. The system of differential equation above derives from it.



For those who are interested we recommend the following web page: http://www.cmp.caltech.edu/~mcc/Chaos_Course/Chua/Chua.html